```
In[1]:= ClearAll["Gloabal`*"];
```

Test 2 Solution

Problem 1

The differential cross section obtained with the Born approximation is in general given by

$$\frac{d\sigma}{d\Theta} = \left(\frac{m}{2\pi\hbar^2}\right)^2 \left|\tilde{V}(q)\right|^2 \tag{1}$$

where $q = \left|\vec{k} - \vec{k}'\right| = 2k\sin\theta$ and $\tilde{V}(q)$ is the Fourier transform of the scattering potential:

$$\tilde{V}(q) = \iiint V(r)\, e^{i\vec{q}\cdot\vec{r}}\, r^2 \sin\theta\, dr\, d\theta\, d\phi \tag{2}$$

$$= 2\pi\left(\frac{\hbar\lambda}{2m}\right)\int_0^\infty dr\,\delta(r-R)\, r^2 \int_0^\pi \sin\theta\, e^{iqr\cos\theta}\, d\theta$$

Integration over θ

```
In[2]:= J1 = Assuming[r > 0 && q > 0, Integrate[Sin[θ] Exp[I q r Cos[θ]], {θ, 0, π}]]
```

Out[2]= $\frac{2\,\text{Sin}[q\,r]}{q\,r}$

Integration over r

```
In[3]:= J2 = Assuming[q > 0 && R > 0, Integrate[DiracDelta[r - R] r^2 J1, {r, 0, Infinity}]]
```

Out[3]= $\frac{2\,R\,\text{Sin}[q\,R]}{q}$

The scattering potential in Fourier space

In[4]:= $Vq = 2\pi\left(\frac{\hbar\lambda}{2m}\right) J2$ // Simplify

Out[4]= $\frac{2\pi R\lambda\hbar\,\text{Sin}[q\,R]}{m\,q}$

In[5]:= $d\sigma = \left(\frac{m}{2\pi\hbar^2}\right)^2 Vq^2$ /. $q \to 2\,k\,\text{Sin}[\theta/2]$

Out[5]= $\frac{R^2\lambda^2\,\text{Csc}\left[\frac{\theta}{2}\right]^2 \text{Sin}\left[2\,k\,R\,\text{Sin}\left[\frac{\theta}{2}\right]\right]^2}{4\,k^2\hbar^2}$

The differential cross section

$$\frac{d\sigma}{d\theta} = \left(\frac{R\lambda}{2k\hbar}\right)^2 \left(\frac{\sin\left[2kR\sin(\theta/2)\right]}{\sin(\theta/2)}\right)^2 \tag{3}$$

Low energy limit

```
In[6]:= Series[dσ, {R, 0, 6}]
```

Out[6]= $\frac{\lambda^2 R^4}{\hbar^2} - \frac{4\left(k^2\lambda^2\,\text{Sin}\left[\frac{\theta}{2}\right]^2\right) R^6}{3\hbar^2} + O[R]^7$

Hence, s-wave scattering cross section is

$$\frac{d\sigma}{d\theta} \approx \frac{\lambda^2 R^4}{\hbar^2} \tag{4}$$

Integrating over θ, we find the total cross section

```
In[7]:= 2 π * Assuming[R > 0 && k > 0, Integrate[dσ Sin[θ], {θ, 0, π}]]
```

$$\text{Out[7]=} \frac{\pi R^2 \lambda^2 (\text{EulerGamma} - \text{CosIntegral}[4 k R] + \text{Log}[4 k R])}{k^2 \hbar^2}$$

$$\sigma = \left(\frac{R \lambda}{k \hbar}\right)^2 \pi [\gamma - \text{Ci}(4 k R) + \ln(4 k R)]$$

$$= \left(\frac{R \lambda}{k \hbar}\right)^2 \pi \, \text{Cin}(4 k R) \tag{5}$$

where Cin(x) is a cosine integral function defined by

$$\text{Cin}(x) = \int_0^x \frac{\cos(t)}{t} dt \tag{6}$$

```
In[8]:= Assuming[x > 0, Series[Integrate[Cos[t] / t, {t, x, Infinity}], {x, 0, 3}]]
```

$$\text{Out[8]=} (-\text{EulerGamma} - \text{Log}[x]) + \frac{x^2}{4} + O[x]^4$$

Hence,

$$\sigma = \left(\frac{R \lambda}{k \hbar}\right)^2 \pi \frac{(4 k R)^2}{4} = \frac{4 \pi \lambda^2 R^4}{\hbar^2} \tag{7}$$

which is consistent with Eq. (4).

Problem 2

```
In[9]:= ψR[x_] = 1/(σ^2 π)^(1/4) Exp[-(x - a)^2 / (2 σ^2)];

In[10]:= ψL[x_] = 1/(σ^2 π)^(1/4) Exp[-(x + a)^2 / (2 σ^2)];
```

(a)

Unnormalized symmetric state is

```
In[11]:= ψB[x, y] = ψL[x] ψR[y] + ψL[y] ψR[x];
```

where x and y are the position of 1st and 2nd particle. The norm of the state is

```
In[12]:= ZB = Sqrt[Assuming[σ > 0 && a > 0,
    Integrate[Integrate[ψB[x, y]^2, {y, -Infinity, Infinity}], {x, -Infinity, Infinity}]]]
```

$$\text{Out[12]=} \sqrt{2} \sqrt{1 + e^{-\frac{2 a^2}{\sigma^2}}}$$

Hence, the normalized symmetric wavefunction is

$$\psi_B(x, y) = [\psi_L(x) \psi_R(y) + \psi_R(y) \psi_L(x)] \Big/ \sqrt{2 \left(1 + e^{-2 a^2/\sigma^2}\right)} \tag{8}$$

```
In[13]:= ψB[x, y] = (ψL[x] ψR[y] + ψL[y] ψR[x]) / ZB;
```

(b) Similarly the antisymmetric state is

In[14]:=
```
ψF[x, y] = ψL[x] ψR[y] - ψL[y] ψR[x];
```

and its norm is

In[15]:=
```
ZF = Sqrt[Assuming[σ > 0 && a > 0,
   Integrate[Integrate[ψF[x, y]^2, {y, -Infinity, Infinity}], {x, -Infinity, Infinity}]]]
```

Out[15]= $\sqrt{2 - 2 e^{-\frac{2 a^2}{\sigma^2}}}$

The normalized antisymmetric wavefunction is

$$\psi_F(x, y) = [\psi_L(x)\,\psi_R(y) - \psi_R(y)\,\psi_L(x)] \Big/ \sqrt{2\left(1 - e^{-2 a^2/\sigma^2}\right)} \tag{9}$$

In[16]:=
```
ψF[x, y] = (ψL[x] ψR[y] - ψL[y] ψR[x]) / ZF;
```

(c)

For boson,

In[17]:=
```
EB = -(ħ^2/(2 m)) Assuming[σ > 0 && a > 0,
    Integrate[Integrate[ψB[x, y] (D[ψB[x, y], x, x] + D[ψB[x, y], y, y]),
      {y, -Infinity, Infinity}], {x, -Infinity, Infinity}]] // FullSimplify
```

Out[17]= $\frac{\hbar^2 \left(-a^2 + \sigma^2 + a^2 \,\mathrm{Tanh}\left[\frac{a^2}{\sigma^2}\right]\right)}{2\, m\, \sigma^4}$

In[18]:=
```
TrigToExp[EB] // Simplify
```

Out[18]= $\frac{\left(-2 a^2 + \left(1 + e^{\frac{2 a^2}{\sigma^2}}\right) \sigma^2\right) \hbar^2}{2 \left(1 + e^{\frac{2 a^2}{\sigma^2}}\right) m\, \sigma^4}$

Simplifying it, we obtain

$$E_B = \left(\frac{\hbar^2}{2 m \sigma^2}\right) \frac{\left(-2 r^2 + 1 + e^{2 r^2}\right)}{1 + e^{2 r^2}} \tag{10}$$

where $r = a/\sigma$.

For fermion,

In[19]:=
```
EF = -(ħ^2/(2 m)) Assuming[σ > 0 && a > 0,
    Integrate[Integrate[ψF[x, y] (D[ψF[x, y], x, x] + D[ψF[x, y], y, y]),
      {y, -Infinity, Infinity}], {x, -Infinity, Infinity}]] // FullSimplify
```

Out[19]= $\frac{\hbar^2 \left(-a^2 + \sigma^2 + a^2 \,\mathrm{Coth}\left[\frac{a^2}{\sigma^2}\right]\right)}{2\, m\, \sigma^4}$

In[20]:=
```
TrigToExp[EF] // Simplify
```

Out[20]= $\frac{\left(2 a^2 + \left(-1 + e^{\frac{2 a^2}{\sigma^2}}\right) \sigma^2\right) \hbar^2}{2 \left(-1 + e^{\frac{2 a^2}{\sigma^2}}\right) m\, \sigma^4}$

$$E_F = \left(\frac{\hbar^2}{2\,m\,\sigma^2}\right)\frac{\left(2\,r^2 - 1 + e^{2\,r^2}\right)}{-1 + e^{2\,r^2}} \tag{11}$$

(d)

For boson,

In[21]:=
```
FB = -D[EB, a] // FullSimplify
```

Out[21]=
$$-\frac{a\,\hbar^2\left(a^2\,\text{Sech}\left[\frac{a^2}{\sigma^2}\right]^2 + \sigma^2\left(-1 + \text{Tanh}\left[\frac{a^2}{\sigma^2}\right]\right)\right)}{m\,\sigma^6}$$

In[22]:=
```
TrigToExp[FB] // FullSimplify
```

Out[22]=
$$\frac{2\,a\left(\sigma^2 + e^{\frac{2a^2}{\sigma^2}}\left(-2\,a^2 + \sigma^2\right)\right)\hbar^2}{\left(1 + e^{\frac{2a^2}{\sigma^2}}\right)^2 m\,\sigma^6}$$

$$F_B = \frac{2}{\sigma}\left(\frac{\hbar^2}{m\,\sigma^2}\right)\frac{r\left(1 - e^{2\,r^2}\left(2\,r^2 - 1\right)\right)}{\left(1 + e^{2\,r^2}\right)^2} \tag{12}$$

For fermion,

In[23]:=
```
FF = -D[EF, a] // FullSimplify
```

Out[23]=
$$\frac{a\,\hbar^2\left(\sigma^2 - \sigma^2\,\text{Coth}\left[\frac{a^2}{\sigma^2}\right] + a^2\,\text{Csch}\left[\frac{a^2}{\sigma^2}\right]^2\right)}{m\,\sigma^6}$$

In[24]:=
```
TrigToExp[FF] // FullSimplify
```

Out[24]=
$$\frac{2\,a\left(\sigma^2 + e^{\frac{2a^2}{\sigma^2}}\left(2\,a^2 - \sigma^2\right)\right)\hbar^2}{\left(-1 + e^{\frac{2a^2}{\sigma^2}}\right)^2 m\,\sigma^6}$$

$$F_F = \frac{2}{\sigma}\left(\frac{\hbar^2}{m\,\sigma^2}\right)\frac{r\left(1 + e^{2\,r^2}\left(2\,r^2 - 1\right)\right)}{\left(1 - e^{2\,r^2}\right)^2} \tag{13}$$

(e)

$$\sigma^2 + e^{\frac{2a^2}{\sigma^2}}\left(2\,a^2 - \sigma^2\right) > \sigma^2 + \left(2\,a^2 - \sigma^2\right) = 2\,a^2 > 0 \tag{14}$$

Hence, $F_F > 0$.

(f)

When $a \gg \sigma$,

$$\sigma^2 + e^{\frac{2a^2}{\sigma^2}}\left(-2\,a^2 + \sigma^2\right) \to e^{\frac{2a^2}{\sigma^2}}\left(-2\,a^2\right) < 0 \tag{15}$$

Hence, $F_B < 0$ for $a \gg \sigma$.